

# Pict: Functional Pictures

Version 6.2.1

August 10, 2015


```
(require pict)    package: pict-lib
```

The `pict` library is one of the standard Racket functional picture libraries (the other being [2htdp/image](#)). This library was originally designed for use with Slideshow, and is re-provided by the `slideshow` language.

# 1 Pict Datatype

A *pict* is a `pict` structure representing an image. Some functions, such as `hline`, create new simple *picts*. Other functions, such as `ht-append`, build new *picts* out of existing *picts*. In the latter case, the embedded *picts* retain their identity, so that offset-finding functions, such as `lt-find`, can find the offset of an embedded *pict* in a larger *pict*.

In addition to its drawing part, a *pict* has the following *bounding box* structure:



That is, the bounding box has a width  $w$  and a height  $h$ . For a single text line,  $d$  is descent below the baseline, and  $a+d=h$ . For multiple text lines (often created with a function like `vc-append`),  $a$  is the ascent of the top line, and  $d$  is the descent of the bottom line, so  $a+d < h$ . Many *picts* have  $d=0$  and  $a=h$ .

In addition, a *pict* can have a *last* sub-*pict* that corresponds to the last item on the last line of text, so that extra lines can be added to the last line. In particular, the *last* element is useful for adding closing parentheses to a block of Racket code, where the last line of code not the longest line in the block.

The size information for a *pict* is computed when the *pict* is created. This strategy supports programs that create new *picts* through arbitrarily complex computations on the size and shape of existing *picts*. The functions `pict-width`, `pict-height`, `pict-descent`, and `pict-ascent` extract bounding box information from a *pict*.

A *pict* is a convertible datatype through the `file/convertible` protocol. Supported conversions include `'png-bytes`, `'eps-bytes`, `'pdf-bytes`, `'svg-bytes`, and variants such as `'png-bytes+bounds` and `'png-bytes+bounds8`.

A *pict* is serializable via `racket/serialize`, but serialization loses sub-*pict* information (preserving only the *pict*'s drawing and bounding box).

Changed in version 1.2 of package `pict-lib`: Added support for `'png-bytes+bounds`, `'png-bytes+bounds8` and similar variants. Changed in version 1.3: Enabled serialization.

```

(struct pict (draw
             width
             height
             ascent
             descent
             children
             panbox
             last)
  #:extra-constructor-name make-pict)
draw : any/c
width : real?
height : real?
ascent : real?
descent : real?
children : (listof child?)
panbox : (or/c #f any/c)
last : (or/c #f pict-path?)

```

A `pict` structure is normally not created directly with `make-pict`. Instead, functions like `text`, `hline`, and `dc` are used to construct a pict.

The `draw` field contains the pict's drawing information in an internal format. Roughly, the drawing information is a procedure that takes a `dc<*>` drawing context and an offset for the pict's top-left corner (i.e., its bounding box's top left corner relative to the `dc<*>` origin). The state of the `dc<*>` is intended to affect the pict's drawing; for example, the pen and brush will be set for a suitable default drawing mode, and the `dc<*>` scale will be set to scale the resulting image. Use `draw-pict` (as opposed to `pict-draw`) to draw the picture.

The `panbox` field is internal and initialized to `#f`.

The `last` field indicates a pict within the `children` list (transitively) that can be treated as the last element of the last line in the pict. A `#f` value means that the pict is its own last sub-pict.

```

(struct child (pict dx dy sx sy sxy syx)
  #:extra-constructor-name make-child)
pict : pict?
dx : real?
dy : real?
sx : real?
sy : real?
sxy : real?
syx : real?

```

Records, for a pict constructed of other pict's, the transformation to arrive at a inverted point in the composed pict from an inverted point in a constituent pict's. An *inverted point* is a

point relative to a pict's lower-left corner with an increasing value moving upward.

A `child` structure is normally not created directly with `make-child`. Instead, functions like `hc-append` create `child` structures when combining pict's to create a new one.

## 2 Basic Pict Constructors

```
(dc draw w h [a d]) → pict?  
draw : (-> (is-a?/c dc<%>) real? real? any)  
w : real?  
h : real?  
a : real? = h  
d : real? = 0
```


Creates an arbitrary self-rendering pict. The arguments to the rendering procedure will be a drawing context and top-left location for drawing.

The *w*, *h*, *a*, and *d* arguments determine the width, height, ascent, and descent of the of the resulting pict's bounding box respectively.

When the rendering procedure is called, the current pen and brush will be 'solid and in the pict's color and linewidth, and the scale and offset of the drawing context will be set. The text mode will be transparent, but the font and text colors are not guaranteed to be anything in particular.

Example:

```
> (dc (λ (dc dx dy)  
      (define old-brush (send dc get-brush))  
      (define old-pen (send dc get-pen))  
      (send dc set-brush  
            (new brush% [style 'fdiagonal-hatch]  
                      [color "darkslategray"])))  
      (send dc set-pen  
            (new pen% [width 3] [color "slategray"])))  
      (define path (new dc-path%))  
      (send path move-to 0 0)  
      (send path line-to 50 0)  
      (send path line-to 25 50)  
      (send path close)  
      (send dc draw-path path dx dy)  
      (send dc set-brush old-brush)  
      (send dc set-pen old-pen))
```



The *draw* is called during the dynamic extent of the call to *dc* as part of the contract checking.

Specifically, the pre-condition portion of the contract for `dc` concocts a `dc< %>` object with a random initial state, calls the `draw` argument with that `dc< %>` and then checks to make sure that `draw` the state of the `dc< %>` object is the same as it was before `draw` was called.

Example:

```
> (dc (λ (dc dx dy)
      (send dc set-brush "red" 'solid)
      (send dc set-pen "black" 1 'transparent)
      (send dc draw-ellipse dx dy 50 50))
    50 50)
```

*dc: contract violation*

*#:pre condition violation; variables are:*

*draw: #<procedure>*

*in: (->i*

*((draw*  
 *(-> (is-a?/c dc< %>) real? real? any))*

*(w real?)*

*(h real?)*

*((d (or/c #f real?)) (a (or/c #f real?)))*

*#:pre*

*(draw)*

*(... draw)*

*(p pict?)*

*contract from: <pkgs>/pict-lib/pict/main.rkt*

*blaming: top-level*

*(assuming the contract is correct)*

*at: <pkgs>/pict-lib/pict/main.rkt:40.3*

Changed in version 1.3 of package `pict-lib`: The `draw` argument is now called by the `#:pre` condition of `dc`.

```
(blank [size]) → pict?
  size : real? = 0
(blank w h) → pict?
  w : real?
  h : real?
(blank w a d) → pict?
  w : real?
  a : real?
  d : real?
(blank w h a d) → pict?
  w : real?
  h : real?
  a : real?
  d : real?
```

Creates a `pict` that draws nothing. The one-argument case supplies a value used for both the

width and height of the resulting pict's bounding box. In the one- and two-argument case, the ascent and descent are 0 for the resulting pict's bounding box; in the three-argument case, the height is computed by adding the given ascent and descent.

Example:

```
> (blank 50)
```

```
(text content [style size angle]) → pict?  
  content : string?  
  style   : text-style/c = null  
  size    : (integer-in 1 1024) = 12  
  angle   : real? = 0
```

Creates a pict that draws text. For creating text pict's within a slide presentation, see [t](#). The size of the resulting pict may depend on the value of [dc-for-text-size](#).

The *style* argument must be one of the following:

- `null` — the default, same as `'default`
- a `font%` object
- a font family symbol, such as `'roman` (see `font%`)
- a font face string, such as `"Helvetica"` (see `font%`)
- `(cons str sym)` combining a face string and a font family (in case the face is unavailable; see `font%`)
- `(cons 'bold style)` for a valid *style*
- `(cons 'italic style)`
- `(cons 'subscript style)`
- `(cons 'superscript style)`
- `(cons 'caps style)`
- `(cons 'combine style)` — allows kerning and ligatures (the default, unless the `'modern` family is specified)
- `(cons 'no-combine style)` — renders characters individually
- `(cons 'aligned style)` — enables hinting, which rounds metrics to integers

- `(cons 'unaligned style)` — disables hinting (which is the default), so that metrics are scalable
- `(cons color style)` — where `color` is a `color%` object, `colorizes` the text

If both `'combine` and `'no-combine` are specified, the first one in `style` takes precedence. Similarly, if both `'aligned` and `'unaligned` are specified, the first one in `style` takes precedence. If `'caps` is specified, the `angle` must be zero.

The given `size` is in pixels, but it is ignored if a `font%` object is provided in the text-style.

The `angle` is in radians, and positive values rotate counter-clockwise. For a non-zero `angle`, the resulting `pict`'s bounding box covers the rotated text, and the descent is zero and the ascent is the height.

Examples:

```
> (text "tom collins")
tom collins
> (text "g & t" (cons 'bold 'roman))
g & t
> (text "martini" null 13 (/ pi 2))
martini
```

```
(hline w h [#:segment seg-length]) → pict?
  w : real?
  h : real?
  seg-length : (or/c #f real?) = #f
(vline w h [#:segment seg-length]) → pict?
  w : real?
  h : real?
  seg-length : (or/c #f real?) = #f
```

Straight lines, centered within their bounding boxes.

Examples:

```
> (hline 40 5)
_____
> (vline 5 40 #:segment 5)
|
|
|
|
|
```



```




(frame pict
  [#:segment seg-length
   #:color color
   #:line-width width]) → pict?
pict : pict?
seg-length : (or/c #f real?) = #f
color : (or/c #f string? (is-a?/c color<%>)) = #f
width : (or/c #f real?) = #f

```

Frames a given pict. If the color or line width are provided, the override settings supplied by the context.

Examples:

```

> (frame (circle 30))

> (frame (circle 30) #:segment 5)

> (frame (circle 30) #:color "chartreuse" #:line-width 3)


```

```

(ellipse w h) → pict?
  w : real?
  h : real?
(circle diameter) → pict?
  diameter : real?
(filled-ellipse w
  h
  [#:draw-border? draw-border?]) → pict?
  w : real?
  h : real?
  draw-border? : any/c = #t
(disk diameter [#:draw-border? draw-border?]) → pict?
  diameter : (and/c rational? (not/c negative?))
  draw-border? : any/c = #t

```

Unfilled and filled ellipses.

If *draw-border?* is #f, then the pen is set to be transparent before drawing the ellipse.

Examples:

```

> (ellipse 40 30)

```



```
> (circle 30)
```



```
> (filled-ellipse 30 40)
```



```
> (disk 30)
```



```
(rectangle w h) → pict?  
  w : real?  
  h : real?  
(filled-rectangle w  
                  h  
                  [#:draw-border? draw-border?]) → pict?  
  w : real?  
  h : real?  
  draw-border? : any/c = #t
```

Unfilled and filled rectangles.

If *draw-border?* is #f, then the pen is set to be transparent before drawing the rectangle.

Examples:

```
> (rectangle 50 50)
```



```
> (filled-rectangle 50 80)
```



```
(rounded-rectangle w  
                  h  
                  [corner-radius  
                  #:angle angle]) → pict?  
  w : real?
```

```

h : real?
corner-radius : real? = -0.25
angle : real? = 0
(filled-rounded-rectangle w
                           h
                           [corner-radius
                            #:angle angle
                            #:draw-border? draw-border?]) → pict?

w : real?
h : real?
corner-radius : real? = -0.25
angle : real? = 0
draw-border? : any/c = #t

```


Unfilled and filled rectangles with rounded corners. The *corner-radius* is used to determine how much rounding occurs in the corners. If it is a positive number, then it determines the radius of a circle touching the edges in each corner, and the rounding of the rectangle follow the edge of those circles. If it is a negative number, then the radius of the circles in the corners is the absolute value of the *corner-radius* times the smaller of *width* and *height*.

The *angle* determines how much the rectangle is rotated, in radians.

If *draw-border?* is *#f*, then the pen is set to be transparent before drawing the rectangle.

Examples:

```
> (rounded-rectangle 40 40 -0.3 #:angle (/ pi 4))
```



```
> (filled-rounded-rectangle 50 40)
```



```

(bitmap img) → pict
img : (or/c path-string?
        (is-a?/c bitmap%)
        (is-a?/c image-snip%))

```

A *pict* that display a *bitmap*. When a *path* is provided, the image is loaded with the *'unknown/mask* flag, which means that a mask *bitmap* is generated if the file contains a mask.

If the *bitmap* cannot be loaded, if the given *bitmap%* object is not valid, or if the *bitmap-draft-mode* parameter is set to *#t*, the result *pict* draws the word “*bitmap failed*”.

```




(arrow size radians) → pict?
  size : real?
  radians : real?
(arrowhead size radians) → pict?
  size : real?
  radians : real?

```

Creates an arrow or arrowhead in the specific direction within a *size* by *size* pict. Points on the arrow may extend slightly beyond the bounding box.

Examples:

```

> (arrow 30 0)

> (arrow 30 (/ pi 2))

> (arrowhead 30 0)


```

```

(pip-line dx dy size) → pict?
  dx : real?
  dy : real?
  size : real?
(pip-arrow-line dx dy size) → pict?
  dx : real?
  dy : real?
  size : real?
(pip-arrows-line dx dy size) → pict?
  dx : real?
  dy : real?
  size : real?

```

Creates a line (with some number of arrowheads) as a zero-sized pict suitable for use with [pin-over](#). The 0-sized picture contains the starting point.

The *size* is used for the arrowhead size. Even though [pip-line](#) creates no arrowheads, it accepts the *size* argument for consistency with the other functions.

```

(pin-line pict
  src
  find-src
  dest
  find-dest
  [#:start-angle start-angle
   #:end-angle end-angle
   #:start-pull start-pull
   #:end-pull end-pull
   #:line-width line-width
   #:color color
   #:alpha alpha
   #:style style
   #:under? under?]) → pict?

pict : pict?
src : pict-path?
find-src : (pict? pict-path? . -> . (values real? real?))
dest : pict-path?
find-dest : (pict? pict-path? . -> . (values real? real?))
start-angle : (or/c real? #f) = #f
end-angle : (or/c real? #f) = #f
start-pull : real? = 1/4
end-pull : real? = 1/4
line-width : (or/c #f real?) = #f
color : (or/c #f string? (is-a?/c color%)) = #f
alpha : (real-in 0.0 1.0) = #f
style : (one-of/c 'transparent 'solid 'xor 'hilite
               'dot 'long-dash 'short-dash 'dot-dash
               'xor-dot 'xor-long-dash 'xor-short-dash
               'xor-dot-dash)
       = 'solid
under? : any/c = #f

```

```

(pin-arrow-line arrow-size
                pict
                src
                find-src
                dest
                find-dest
                [#:start-angle start-angle
                #:end-angle end-angle
                #:start-pull start-pull
                #:end-pull end-pull
                #:line-width line-width
                #:color color
                #:alpha alpha
                #:style style
                #:under? under?
                #:solid? solid?
                #:hide-arrowhead? hide-arrowhead?]) → pict?

arrow-size : real?
pict : pict?
src : pict-path?
find-src : (pict? pict-path? . -> . (values real? real?))
dest : pict-path?
find-dest : (pict? pict-path? . -> . (values real? real?))
start-angle : (or/c real? #f) = #f
end-angle : (or/c real? #f) = #f
start-pull : real? = 1/4
end-pull : real? = 1/4
line-width : (or/c #f real?) = #f
color : (or/c #f string? (is-a?/c color%)) = #f
alpha : (real-in 0.0 1.0) = #f
style : (one-of/c 'transparent 'solid 'xor 'hilite
                'dot 'long-dash 'short-dash 'dot-dash
                'xor-dot 'xor-long-dash 'xor-short-dash
                'xor-dot-dash)
        = 'solid
under? : any/c = #f
solid? : any/c = #t
hide-arrowhead? : any/c = #f

```

```

(pin-arrows-line arrow-size
                 pict
                 src
                 find-src
                 dest
                 find-dest
                 [#:start-angle start-angle
                 #:end-angle end-angle
                 #:start-pull start-pull
                 #:end-pull end-pull
                 #:line-width line-width
                 #:color color
                 #:alpha alpha]
                 [#:under? under?
                 #:solid? solid?
                 #:hide-arrowhead? hide-arrowhead?]) → pict?

arrow-size : real?
pict : pict?
src : pict-path?
find-src : (pict? pict-path? . -> . (values real? real?))
dest : pict-path?
find-dest : (pict? pict-path? . -> . (values real? real?))
start-angle : (or/c real? #f) = #f
end-angle : (or/c real? #f) = #f
start-pull : real? = 1/4
end-pull : real? = 1/4
line-width : (or/c #f real?) = #f
color : (or/c #f string? (is-a?/c color%)) = #f
alpha : (real-in 0.0 1.0) = #f
style : (one-of/c 'transparent 'solid 'xor 'hilite
                'dot 'long-dash 'short-dash 'dot-dash
                'xor-dot 'xor-long-dash 'xor-short-dash
                'xor-dot-dash)
under? : any/c = #f
solid? : any/c = #t
hide-arrowhead? : any/c = #f

```

Adds a line or line-with-arrows onto *pict*, using one of the pict-finding functions (e.g., *lt-find*) to extract the source and destination of the line.

If *under?* is true, then the line and arrows are added under the existing *pict* drawing, instead of on top. If *solid?* is false, then the arrowheads are hollow instead of filled.

The *start-angle*, *end-angle*, *start-pull*, and *end-pull* arguments control the curve of the line (and the defaults produce a straight line):


- The *start-angle* and *end-angle* arguments specify the direction of curve at its start and end positions; if either is *#f*, it defaults to the angle of a straight line from the start position to end position.
- The *start-pull* and *end-pull* arguments specify a kind of momentum for the starting and ending angles; larger values preserve the angle longer.

The *line-width*, *color*, *alpha*, and *style* arguments apply to the added line.

When the *hide-arrowhead?* argument is a true value, then space for an arrowhead is kept around the line, but the arrowhead itself is not drawn.

Examples:

```
(define pict-a (rectangle 40 40))
(define pict-b (circle 40))
(define combined (hc-append 200 pict-a pict-b))
> (pin-line combined
    pict-a cc-find
    pict-b cc-find)
```



```
> (pin-arrow-line 30 combined
    pict-a rc-find
    pict-b lc-find
    #:line-width 3
    #:style 'long-dash
    #:color "medium goldenrod")
```



```
> (pin-arrows-line 30 combined
    pict-a rc-find
    pict-b lc-find
    #:start-angle (/ pi 11)
    #:end-angle (- (/ pi 11))
    #:solid? #f)
```



`text-style/c` : contract?

A contract that matches the second argument of `text`.



```
(bitmap-draft-mode) → boolean?  
(bitmap-draft-mode on?) → void?  
  on? : any/c
```

A parameter that determines whether `bitmap` loads/uses a bitmap.

### 3 Pict Combiners

```
(vl-append [d] pict ...) → pict?  
  d : real? = 0.0  
  pict : pict?  
(vc-append [d] pict ...) → pict?  
  d : real? = 0.0  
  pict : pict?  
(vr-append [d] pict ...) → pict?  
  d : real? = 0.0  
  pict : pict?  
(ht-append [d] pict ...) → pict?  
  d : real? = 0.0  
  pict : pict?  
(htl-append [d] pict ...) → pict?  
  d : real? = 0.0  
  pict : pict?  
(hc-append [d] pict ...) → pict?  
  d : real? = 0.0  
  pict : pict?  
(hbl-append [d] pict ...) → pict?  
  d : real? = 0.0  
  pict : pict?  
(hb-append [d] pict ...) → pict?  
  d : real? = 0.0  
  pict : pict?
```

Creates a new pict as a column (for `v...-append`) or row (for `h...-append`) of other pict. The optional `d` argument specifies amount of space to insert between each pair of pictures in making the column or row.

Different procedures align pictures in the orthogonal direction in different ways. For example, `vl-append` left-aligns all of the pictures.

The descent of the result corresponds to baseline that is lowest in the result among all of the pict's descent-specified baselines; similarly, the ascent of the result corresponds to the highest ascent-specified baseline. If at least one `pict` is supplied, then the last element (as reported by `pict-last`) for the result is (or `(pict-last pict) pict`) for the using last supplied `pict`.

Examples:

```
(define combiners (list vl-append vc-append vr-append  
                        ht-append htl-append hc-append  
                        hbl-append hb-append))
```

```

(define names (list "vl-append" "vc-append" "vr-append"
                  "ht-append" "htl-append" "hc-append"
                  "hbl-append" "hb-append"))

(define pict-a (colorize (filled-rectangle 60 30) "tomato"))

(define pict-b (colorize (disk 45) "cornflower blue"))

(define pict
  (for/list ([combiner combinators] [name names])
    (list (text name null 15)
          (combiner pict-a pict-b))))

```

```
> (take pict 4)
```

```

      
      '((vl-append  ) (vc-append  )
      (vr-append  ) (ht-append  ))

```

```
> (drop pict 4)
```

```

      
      '((htl-append   ) (hc-append   )
      (hbl-append   ) (hb-append   ))

```

```

(lt-superimpose pict ...) → pict?
 pict : pict?
(ltl-superimpose pict ...) → pict?
 pict : pict?
(lc-superimpose pict ...) → pict?
 pict : pict?
(lbl-superimpose pict ...) → pict?
 pict : pict?
(lb-superimpose pict ...) → pict?
 pict : pict?
(ct-superimpose pict ...) → pict?
 pict : pict?
(ctl-superimpose pict ...) → pict?

```

```

  pict : pict?
(cc-superimpose pict ...) → pict?
  pict : pict?
(cbl-superimpose pict ...) → pict?
  pict : pict?
(cb-superimpose pict ...) → pict?
  pict : pict?
(rt-superimpose pict ...) → pict?
  pict : pict?
(rtl-superimpose pict ...) → pict?
  pict : pict?
(rc-superimpose pict ...) → pict?
  pict : pict?
(rbl-superimpose pict ...) → pict?
  pict : pict?
(rb-superimpose pict ...) → pict?
  pict : pict?

```

Creates a new picture by superimposing a set of pictures. The name prefixes are alignment indicators: horizontal alignment then vertical alignment.

The descent of the result corresponds to baseline that is lowest in the result among all of the *picts*' descent-specified baselines; similarly, the ascent of the result corresponds to the highest ascent-specified baseline. The last element (as reported by `pict-last`) for the result is the lowest, right-most among the last-element *picts* of the *pict* arguments, as determined by comparing the last-element bottom-right corners.

Examples:

```

(define combiners (list lt-superimpose ltl-superimpose lc-
  superimpose
                        lbl-superimpose lb-superimpose ct-
superimpose
                        ctl-superimpose cc-superimpose cbl-
superimpose
                        cb-superimpose rt-superimpose rtl-
superimpose
                        rc-superimpose rbl-superimpose rb-
superimpose))

(define names (list "lt-superimpose" "ltl-superimpose" "lc-
superimpose"
                   "lbl-superimpose" "lb-superimpose" "ct-
superimpose"
                   "ctl-superimpose" "cc-superimpose" "cbl-
superimpose"

```

```

"cb-superimpose" "rt-superimpose" "rtl-
superimpose"
"rc-superimpose" "rbl-superimpose" "rb-
superimpose"))

```

```

(define pict-a (colorize (filled-rectangle 60 30) "tomato"))

```

```

(define pict-b (colorize (disk 45) "cornflower blue"))

```

```

(define picts
  (for/list ([combiner combiners] [name names])
    (list (text name null 15)
          (combiner pict-a pict-b))))



```

```

> (take picts 3)

```

```

'((ltsuperimpose  ) (ltsuperimpose  ))

```

```

(lcsuperimpose  ))

```

```

> (take (drop picts 3) 3)


```

```

'((lbsuperimpose  ) (lbsuperimpose  ))

```

```

(ctsuperimpose  ))



```

```

> (take (drop picts 6) 3)


```

```

'((ctlsuperimpose  ) (ccsuperimpose  ))

```

```

(cblsuperimpose  ))



```

```

> (take (drop picts 9) 3)


```

```

'((cbsuperimpose  ) (rtsuperimpose  ))

```

```

(rtlsuperimpose  ))

```

```

> (take (drop picts 12) 3)

```

```
'((rc-superimpose (img "circle" 100 100) (img "square" 100 100)) (rbl-superimpose (img "circle" 100 100) (img "square" 100 100))
  (rb-superimpose (img "circle" 100 100) (img "square" 100 100)))
```

```
(pin-over base dx dy pict) → pict?
  base : pict?
  dx : real?
  dy : real?
  pict : pict?
(pin-over base find-pict find pict) → pict?
  base : pict?
  find-pict : pict-path?
  find : (pict? pict-path? . -> . (values real? real?))
  pict : pict?
```

Creates a pict with the same bounding box, ascent, and descent as *base*, but with *pict* placed on top. The *dx* and *dy* arguments specify how far right and down the second pict's corner is from the first pict's corner. Alternately, the *find-pict* and *find* arguments find a point in *base* for *find-pict*; the *find* procedure should be something like `lt-find`.

```
(pin-under base dx dy pict) → pict?
  base : pict?
  dx : real?
  dy : real?
  pict : pict?
(pin-under base find-pict find pict) → pict?
  base : pict?
  find-pict : pict?
  find : (pict? pict? . -> . (values real? real?))
  pict : pict?
```

Like `pin-over`, but *pict* is drawn before *base* in the resulting combination.

```
(table ncols
      picts
      col-aligns
      row-aligns
      col-seps
      row-seps) → pict?
ncols : exact-positive-integer?
picts : (non-empty-listof pict?)
col-aligns : (or/c (list*of (->* () #:rest (listof pict?) pict?))
                  (listof (->* () #:rest (listof pict?) pict?)))
```

```

row-aligns : (or/c (list*of (->* () #:rest (listof pict?) pict?))
                  (listof (->* () #:rest (listof pict?) pict?)))
col-seps : (or/c (list*of real?) (listof real?))
row-seps : (or/c (list*of real?) (listof real?))

```

Creates a table given a list of pict's. The *picts* list is a concatenation of the table's rows (which means that a Racket `list` call can be formatted to reflect the shape of the output table).

The *col-aligns*, *row-aligns*, *col-seps*, and *row-seps* arguments are "lists" specifying the row and columns alignments separation between rows and columns. For *c* columns and *r* rows, the first two should have *c* and *r* superimpose procedures, and the last two should have *c*-1 and *r*-1 numbers, respectively. The lists can be "improper" (i.e., ending in a number instead of an empty list), in which case the non-pair `cdr` is used as the value for all remaining list items that were expected. The *col-aligns* and *row-aligns* procedures are used to superimpose all of the cells in a column or row; this superimposition determines the total width or height of the column or row, and also determines the horizontal or vertical placement of each cell in the column or row.

Examples:

```

> (table 4
      (map (λ (x) (text (format "~a" x)))
            (list 1 2 3 4
                  5 6 7 8
                  9000 10 11 12)))
      cc-superimpose
      cc-superimpose
      10
      10)
1    2    3    4
5    6    7    8
9000 10 11 12

> (table 4
      (map (λ (x) (text (format "~a" x)))
            (list 1 2 3 4
                  5 6 7 8
                  9000 10 11 12)))
      rc-superimpose
      cc-superimpose
      10
      10)

```

1 2 3 4  
5 6 7 8  
9000 10 11 12






## 4 Pict Drawing Adjusters

```
(scale pict factor) → pict?  
  pict : pict?  
  factor : real?  
(scale pict w-factor h-factor) → pict?  
  pict : pict?  
  w-factor : real?  
  h-factor : real?
```

Scales a pict drawing, as well as its bounding box, by multiplying its current size by *factor* (if two arguments are supplied) or by multiplying the current width by *w-factor* and current height by *h-factor* (if three arguments are supplied).

The drawing is scaled by adjusting the destination `dc<%>`'s scale while drawing the original *pict*.

Examples:

```
> (filled-rectangle 80 40)  
  
> (scale (filled-rectangle 40 20) 2)  
  
> (scale (filled-rectangle 20 20) 4 2)  

```

```
(scale-to-fit pict size-pict) → pict?  
  pict : pict?  
  size-pict : pict?  
(scale-to-fit pict width height) → pict?  
  pict : pict?  
  width : real?  
  height : real?
```

Scales *pict* so that it fits within the bounding box of *size-pict* (if two arguments are supplied) or into a box of size *width* by *height* (if three arguments are supplied).

The aspect ratio of the pict is preserved, so the resulting pict will have either the width or the height of the *size-pict* (or *width* by *height* box), but not necessarily both.

```
(rotate pict theta) → pict?
  pict : pict?
  theta : real?
```

Rotates a pict’s drawing by *theta* radians counter-clockwise.

The bounding box of the resulting pict is the box encloses the rotated corners of *pict* (which inflates the area of the bounding box, unless *theta* is a multiple of half of *pi*). The ascent and descent lines of the result’s bounding box are the horizontal lines that bisect the rotated original lines; if the ascent line drops below the descent line, the two lines are flipped.

```
(ghost pict) → pict?
  pict : pict?
```

Creates a container picture that doesn’t draw the child picture, but uses the child’s size.

```
(linewidth w pict) → pict?
  w : (or/c real? #f)
  pict : pict?
```

Selects a specific pen width for drawing, which applies to pen drawing for *pict* that does not already use a specific pen width. A *#f* value for *w* makes the pen transparent (in contrast to a zero value, which means “as thin as possible for the target device”).

```
(linestyle style pict) → pict?
  style : (one-of/c 'transparent 'solid 'xor 'hilite
                  'dot 'long-dash 'short-dash 'dot-dash
                  'xor-dot 'xor-long-dash 'xor-short-dash
                  'xor-dot-dash)
  pict : pict?
```

Selects a specific pen style for drawing, which applies to pen drawing for *pict* that does not already use a specific pen style.

```
(colorize pict color) → pict?
  pict : pict?
  color : (or/c string? (is-a?/c color%)
          (list/c byte? byte? byte?))
```

Selects a specific color drawing, which applies to drawing in *pict* that does not already use a specific color. The **black-and-white** parameter causes all non-white colors to be converted to black.

```
(cellophane pict opacity) → pict?
  pict : pict?
  opacity : (real-in 0 1)
```

Makes the given *pict* semi-transparent, where an opacity of 0 is fully transparent, and an opacity of 1 is fully opaque. See [set-alpha](#) for information about the contexts and cases when semi-transparent drawing works.

```
(clip pict) → pict  
pict : pict?
```

Clips a *pict*'s drawing to its bounding box.

```
(inset/clip pict amt) → pict?  
pict : pict?  
amt : real?  
(inset/clip pict h-amt v-amt) → pict?  
pict : pict?  
h-amt : real?  
v-amt : real?  
(inset/clip pict l-amt t-amt r-amt b-amt) → pict?  
pict : pict?  
l-amt : real?  
t-amt : real?  
r-amt : real?  
b-amt : real?
```

Insets and clips the *pict*'s drawing to its bounding box. Usually, the inset amounts are negative.

```
(scale/improve-new-text pict-expr scale-expr)  
(scale/improve-new-text pict-expr x-scale-expr y-scale-expr)
```

Like the [scale](#) procedure, but also sets [current-expected-text-scale](#) while evaluating *pict-expr*.

```
(black-and-white) → boolean?  
(black-and-white on?) → void?  
on? : any/c
```

A parameter that determines whether [colorize](#) uses color or black-and-white colors.

## 5 Bounding Box Adjusters

```
(inset pict amt) → pict?  
  pict : pict?  
  amt : real?  
(inset pict h-amt v-amt) → pict?  
  pict : pict?  
  h-amt : real?  
  v-amt : real?  
(inset pict l-amt t-amt r-amt b-amt) → pict?  
  pict : pict?  
  l-amt : real?  
  t-amt : real?  
  r-amt : real?  
  b-amt : real?
```

Extends *pict*'s bounding box by adding the given amounts to the corresponding sides; ascent and descent are extended, too.

```
(clip-descent pict) → pict?  
  pict : pict?
```

Truncates *pict*'s bounding box by removing the descent part.

```
(lift-above-baseline pict amt) → pict?  
  pict : pict?  
  amt : real?
```

Lifts *pict* relative to its baseline, extending the bounding box height if necessary.

```
(drop-below-ascent pict amt) → pict?  
  pict : pict?  
  amt : real?
```

Drops *pict* relative to its ascent line, extending the bounding box height if necessary.

```
(baseless pict) → pict?  
  pict : pict?
```

Makes the descent 0 and the ascent the same as the height.

```
(refocus pict sub-pict) → pict?  
  pict : pict?  
  sub-pict : pict?
```

Assuming that *sub-pict* can be found within *pict*, shifts the overall bounding box to that of *sub-pict* (but preserving all the drawing of *pict*). The last element, as reported by `pict-last` is also set to `(or (pict-last sub-pict) sub-pict)`.

```
(panorama pict) → pict?  
pict : pict?
```

Shifts the given *pict*'s bounding box to enclose the bounding boxes of all sub-picts (even `laundered` pict).

```
(use-last pict sub-pict) → pict?  
pict : pict?  
sub-pict : pict-path?
```

Returns a *pict* like *pict*, but with the last element (as reported by `pict-last`) set to *sub-pict*. The *sub-pict* must exist as a sub-pict (or path of sub-picts) within *pict*.

```
(use-last* pict sub-pict) → pict?  
pict : pict?  
sub-pict : pict-path?
```

Propagates the last element of *sub-pict* to *pict*.

That is, `use-last*` is like `use-last`, but the last element of *sub-pict* is used as the new last element for *pict*, instead of *sub-pict* itself—unless `(pict-last sub-pict)` is `#f`, in which case *sub-pict* is used as the last element of *pict*.

## 6 Pict Finders

```
(lt-find pict find) → real? real?  
  pict : pict?  
  find  : pict-path?  
(ltl-find pict find) → real? real?  
  pict : pict?  
  find  : pict-path?  
(lc-find pict find) → real? real?  
  pict : pict?  
  find  : pict-path?  
(lbl-find pict find) → real? real?  
  pict : pict?  
  find  : pict-path?  
(lb-find pict find) → real? real?  
  pict : pict?  
  find  : pict-path?  
(ct-find pict find) → real? real?  
  pict : pict?  
  find  : pict-path?  
(ctl-find pict find) → real? real?  
  pict : pict?  
  find  : pict-path?  
(cc-find pict find) → real? real?  
  pict : pict?  
  find  : pict-path?  
(cbl-find pict find) → real? real?  
  pict : pict?  
  find  : pict-path?  
(cb-find pict find) → real? real?  
  pict : pict?  
  find  : pict-path?  
(rt-find pict find) → real? real?  
  pict : pict?  
  find  : pict-path?  
(rtl-find pict find) → real? real?  
  pict : pict?  
  find  : pict-path?  
(rc-find pict find) → real? real?  
  pict : pict?  
  find  : pict-path?  
(rbl-find pict find) → real? real?  
  pict : pict?  
  find  : pict-path?  
(rb-find pict find) → real? real?
```

```
  pict : pict?  
  find : pict-path?
```

Locates a pict designated by *find* is within *pict*. If *find* is a pict, then the *pict* must have been created as some combination involving *find*.

If *find* is a list, then the first element of *find* must be within *pict*, the second element of *find* must be within the second element, and so on.

```
(pict-path? v) → boolean?  
  v : any/c
```

Returns `#t` if *v* is a *pict* or a non-empty list of *picts*.

```
(launder pict) → pict?  
  pict : pict?
```

Creates a pict that has the same drawing and bounding box of *pict*, but which hides all of its sub-picts so that they cannot be found with functions like `lt-find`. If *pict* has a last-line pict, then the laundered pict has a fresh last-line pict with the same shape and location.

## 7 More Pict Constructors

### 7.1 Dingbats

```
(cloud w h [color]) → pict?  
w : real?  
h : real?  
color : (or/c string? (is-a?/c color%)) = "gray"
```

Creates a fluffy cloud.

Examples:

```
> (cloud 100 75)
```



```
> (cloud 100 75 "lavenderblush")
```




```
(file-icon w h color [shaded?]) → pict?  
w : real?  
h : real?  
color : (or/c string? (is-a?/c color%) any/c)  
shaded? : any/c = #f
```

Creates a Mac-like file icon, optionally shaded. If `color` is not a string or `color%` object, it is treated as a boolean, in which case true means "gray" and false means "white".


Examples:

```
> (file-icon 50 60 "bisque")
```



```
> (file-icon 50 60 "honeydew" #t)
```





```
(standard-fish w
              h
              [#:direction direction
              #:color color
              #:eye-color eye-color
              #:open-mouth open-mouth]) → pict?


w : real?
h : real?
direction : (or/c 'left 'right) = 'left
color : (or/c string? (is-a?/c color%)) = "blue"
eye-color : (or/c string? (is-a?/c color%) #f) = "black"
open-mouth : (or/c boolean? real?) = #f
```

Creates a fish swimming either 'left or 'right. If *eye-color* is #f, no eye is drawn.


The *open-mouth* argument can be either #f (mouth closed), #t (mouth fully open), or a number: 0.0 is closed, 1.0 is fully open, and numbers in between are partially open.

Examples:


```
> (standard-fish 100 50)
```




```
> (standard-fish 100 50 #:direction 'right #:color "chocolate")
```



```
> (standard-fish 100 50 #:eye-color "saddlebrown" #:color "salmon")
```



```
> (standard-fish 100 50 #:open-mouth #t #:color "olive")
```



```
(jack-o-lantern size
               [pumpkin-color
               face-color]) → pict?
```

```
size : real?  
pumpkin-color : (or/c string? (is-a?/c color%)) = "orange"  
face-color : (or/c string? (is-a?/c color%)) = "black"
```

Creates a jack-o-lantern; use the same pumpkin and face color to get a plain pumpkin. The *size* determines the width.

Examples:

```
> (jack-o-lantern 100)
```



```
> (jack-o-lantern 100 "cadet blue" "khaki")
```



```
(angel-wing w h left?) → pict?  
w : real?  
h : real?  
left? : any/c
```

Creates an angel wing, left or right, or any size. The color and pen width for drawing the wing outline is the current one.

Examples:

```
> (angel-wing 100 40 #f)
```



```
> (angel-wing 100 40 #t)
```



```
(desktop-machine scale [style]) → pict?  
  scale : real?  
  style : (listof symbol?) = null
```


Produces a picture of ancient desktop computer. The *scale* argument scales the size relative to the base size of 120 by 115.

The *style* can include any of the following:


- 'plt — include a Racket logo on the machine's screen
- 'binary — put 1s and 0s on the machine's screen
- 'devil — like 'binary, and also give the machine horns and a tail

Examples:


```
> (desktop-machine 1)
```



```
> (desktop-machine 1 '(devil plt))
```



```
> (desktop-machine 1 '(plt binary))
```



```

(thermometer [#:height-% height-%
              #:color-% color-%
              #:ticks ticks
              #:start-color start-color
              #:end-color end-color
              #:top-circle-diameter top-circle-diameter
              #:bottom-circle-diameter bottom-circle-diameter
              #:stem-height stem-height
              #:mercury-inset mercury-inset])
→ pict?
height-% : (between/c 0 1) = 1
color-% : (between/c 0 1) = height-%
ticks : non-exact-negative-integer? = 4
start-color : (or/c string? (is-a?/c color%)) = "lightblue"
end-color : (or/c string? (is-a?/c color%)) = "lightcoral"
top-circle-diameter : positive-real? = 40
bottom-circle-diameter : positive-real? = 80
stem-height : positive-real? = 180
mercury-inset : positive-real? = 8

```

Produces a thermometer that consists of a semi-circle on top of a rectangle on top of a circle. The sizes of the three components are controlled via the *top-circle-diameter*, *stem-height*, and *bottom-circle-diameter* arguments.

The mercury is drawn the same way, but by creating the three components inset from the versions that draw the boundary of the thermometer. This inset is controlled by the *mercury-inset* argument.

The height of the mercury in the thermometer is controlled by the *height-%* argument. Its color is interpolated between the *start-color* and *end-color*, as determined by the *color-%* argument.


Finally, some number of ticks are drawn, based on the *ticks* argument.

Example:

```

> (thermometer #:stem-height 90
              #:bottom-circle-diameter 40
              #:top-circle-diameter 20
              #:mercury-inset 4)

```



## 7.2 Balloon Annotations

```
(require pict/balloon)    package: pict-lib
```

The `pict/balloon` library provides functions for creating and placing cartoon-speech balloons.

```
(wrap-balloon pict
              spike
              dx
              dy
              [color
              corner-radius]) → balloon?
pict : pict?
spike : (or/c 'n 's 'e 'w 'ne 'se 'sw 'nw)
dx : real?
dy : real?
color : (or/c string? (is-a?/c color%)) = balloon-color
corner-radius : (and/c real? (not/c negative?)) = 32
```

Superimposes `pict` on top of a balloon that wraps it.

The `spike` argument indicates the corner from which a spike protrudes from the balloon (i.e., the spike that points to whatever the balloon is about). For example, `'n` means “north,” which is a spike in the top middle of the balloon.

The `dx` and `dy` arguments specify how far the spike should protrude. For a `'w` spike, `dx` should be negative, etc.

The `color` argument is the background color for the balloon.

The `corner-radius` argument determines the radius of the circle used to round the balloon’s corners. As usual, if it is less than `1`, then it acts as a ratio of the balloon’s width or height.

The result is a balloon, not a pict. The `balloon-pict` function extracts a pict whose bounding box does not include the spike, but includes the rest of the image, and the `balloon-`

`point-x` and `balloon-point-y` functions extract the location of the spike point. More typically, the `pin-balloon` function is used to add a balloon to a pict.

```
(pip-wrap-balloon pict
                  spike
                  dx
                  dy
                  [color
                  corner-radius]) → pict?
pict : pict?
spike : (or/c 'n 's 'e 'w 'ne 'se 'sw 'nw)
dx : real?
dy : real?
color : (or/c string? (is-a?/c color%)) = balloon-color
corner-radius : (and/c real? (not/c negative?)) = 32
```

Like `wrap-balloon`, but produces a zero-sized pict suitable for use with `pin-over`.

```
(pin-balloon balloon base x y) → pict?
balloon : balloon?
base : pict?
x : real?
y : real?
(pin-balloon balloon base at-pict find) → pict?
balloon : balloon?
base : pict?
at-pict : pict-path?
find : (pict? pict-path? . -> . (values real? real?))
```


Superimposes the pict in `balloon` onto `base` to produce a new pict. The balloon is positioned so that its spike points to the location specified by either `x` and `y` (numbers) or at the position determined by combining `base` and `at-pict` with `find`. The `find` function uses its arguments like `lt-find`.

The resulting pict has the same bounding box, descent, and ascent as `base`, even if the balloon extends beyond the bounding box.


Examples:

```
> (define a-pict (standard-fish 70 40))

> (pin-balloon (balloon 40 30 5 'se 5 5)
              (cc-superimpose (blank 300 150) a-pict)
              a-pict
              lc-find)
```



```
> (pin-balloon (wrap-balloon (text "Hello!") 'sw -5 3)
      (cc-superimpose (blank 300 150) a-pict)
      a-pict
      rt-find)
```



```
(balloon w h corner-radius spike dx dy [color]) → balloon?
  w : real?
  h : real?
  corner-radius : (and/c real? (not/c negative?))
  spike : (or/c 'n 's 'e 'w 'ne 'se 'sw 'nw)
  dx : real?
  dy : real?
  color : (or/c string? (is-a?/c color%)) = balloon-color
```

Creates a balloon, much like `wrap-balloon` except that the balloon's width is `w` and its height is `h`.

```
(balloon? v) → boolean?
  v : any/c
(make-balloon pict x y) → balloon?
  pict : pict?
  x : real?
  y : real?
(balloon-pict balloon) → pict?
  balloon : balloon?
(balloon-point-x balloon) → real?
```

```

    balloon : balloon?
    (balloon-point-y balloon) → real?
    balloon : balloon?

```

A balloon encapsulates a pict and the position of the balloon's spike relative to the balloon's top-left corner.

```

balloon-color : (or/c string? (is-a?/c color%))

```

The default background color for a balloon.

```

(balloon-enable-3d) → boolean?
(balloon-enable-3d on?) → void?
    on? : any/c

```

A parameter that determines whether balloons are drawn with 3-D shading.

### 7.3 Face

```

(require pict/face)      package: pict-lib

```

The `pict/face` library provides functions for a kind of Mr. Potatohead-style face library.

```

default-face-color : (or/c string (is-a?/c color%))

```

Orange.

```

(face mood [color]) → pict?
    mood : symbol?
    color : (or/c string (is-a?/c color%)) = default-face-color

```

Returns a pict for a pre-configured face with the given base color. The built-in configurations, selected by mood-symbol, are as follows:

```

'unhappy — (face* 'none 'plain #t default-face-color 6)

```

```

'sortof-unhappy — (face* 'worried 'grimace #t default-face-color 6)

```





```
'sortof-happy — (face* 'worried 'medium #f default-face-color 6)
```

```
'happy — (face* 'none 'plain #f default-face-color 6)
```

```
'happier — (face* 'none 'large #f default-face-color 3)
```

```
'embarrassed — (face* 'worried 'medium #f default-face-color 3)
```

```
'badly-embarrassed — (face* 'worried 'medium #t default-face-color 3)
```

```
'unhappier — (face* 'normal 'large #t default-face-color 3)
```

```
'happiest — (face* 'normal 'huge #f default-face-color 0 -3)
```

```
'unhappiest — (face* 'normal 'huge #t default-face-color 0 -3)
```



```
'mad — (face* 'angry 'grimace #t default-face-color 0)
```

```
'mean — (face* 'angry 'narrow #f default-face-color 0)
```

```
'surprised — (face* 'worried 'oh #t default-face-color -4 -3 2)
```

```
(face* eyebrow-kind
      mouth-kind
      frown?
      color
      eye-inset
      eyebrow-dy
      pupil-dx
      pupil-dy
      [#:eyebrow-shading? eyebrow-on?
       #:mouth-shading? mouth-on?
       #:eye-shading? eye-on?
       #:tongue-shading? tongue-on?
       #:face-background-shading? face-bg-on?
       #:teeth? teeth-on?]) → pict?
eyebrow-kind : (or/c 'none 'normal 'worried 'angry)
mouth-kind : (or/c 'plain 'smaller 'narrow 'medium 'large
                  'huge 'grimace 'oh 'tongue)
frown? : any/c
color : (or/c string (is-a?/c color%))
eye-inset : real?
eyebrow-dy : real?
pupil-dx : real?
pupil-dy : real?
eyebrow-on? : any/c = #t
mouth-on? : any/c = #t
eye-on? : any/c = #t
tongue-on? : any/c = #t
face-bg-on? : any/c = #t
teeth-on? : any/c = #t
```



Returns a pict for a face:

- *eyebrow-kind* determines the eyebrow shape.
- *mouth-kind* determines the mouth shape, combined with *frown?*.
- *frown?* determines whether the mouth is up or down.
- *color* determines the face color.
- *eye-inset* adjusts the eye size; recommend values are between 0 and 10.
- *eyebrow-dy* adjusts the eyebrows; recommend values: between -5 and 5.
- *pupil-dx* adjusts the pupil; recommend values are between -10 and 10.
- *pupil-dy* adjusts the pupil; recommend values are between -15 and 15.

The `#:eyebrow-shading?` through `#:face-background-shading?` arguments control whether a shading is used for on a particular feature in the face (shading tends to look worse than just anti-aliasing when the face is small). The `#:teeth?` argument controls the visibility of the teeth for some mouth shapes.

## 7.4 Flash

```
(require pict/flash)      package: pict-lib

(filled-flash width
              height
              [n-points
              spike-fraction
              rotation])  → pict?

width : real?
height : real?
n-points : exact-positive-integer? = 10
spike-fraction : (real-in 0 1) = 0.25
rotation : real? = 0
```

Returns a pict for a “flash”: a spiky oval, like the yellow background that goes behind a “new!” logo on web pages or a box of cereal.

The *height* and *width* arguments determine the size of the oval in which the flash is drawn, prior to rotation. The actual height and width may be smaller if *points* is not a multiple of 4, and the actual height and width will be different if the flash is rotated.

The *n-points* argument determines the number of points on the flash.

The *spike-fraction* argument determines how big the flash spikes are compared to the bounding oval.

The *rotation* argument specifies an angle in radians for counter-clockwise rotation.

The flash is drawn in the default color.

Examples:

```
> (filled-flash 100 50)
```



```
> (filled-flash 100 50 8 0.25 (/ pi 2))
```




```
(outline-flash width
               height
               [n-points
               spike-fraction
               rotation]) → pict?
width : real?
height : real?
n-points : exact-positive-integer? = 10
spike-fraction : (real-in 0 1) = 0.25
rotation : real? = 0
```


Like *filled-flash*, but drawing only the outline.

Examples:

```
> (outline-flash 100 50)
```



```
> (outline-flash 100 50 8 0.25 (/ pi 2))
```



## 7.5 Typesetting Racket Code

```
(require pict/code)    package: pict-lib
```

This library is re-provided by `slideshow/code`, but initializes `get-current-code-font-size`.

```
(typeset-code stx) → pict?  
  stx : syntax?
```

Produces a `pict` for code in the given syntax object. The source-location information of the syntax object determines the line breaks, line indenting, and space within a row. Empty rows are ignored.

Beware that if you use `read-syntax` on a file port, you may have to turn on line counting via `port-count-lines!` for the code to typeset properly. Also beware that when a source file containing a `syntax` or `quote-syntax` form is compiled, source location information is omitted from the compiled syntax object.

Normally, `typeset-code` is used through the `code` syntactic form, which works properly with compilation, and that escapes to `pict`-producing code via `unsyntax`. See also `define-code`.

Embedded `picts` within `stx` are used directly. Row elements are combined using an operator like `htl-append`, so use `code-align` (see below) as necessary to add an ascent to ascentless `picts`. More precisely, creation of a line of code uses `pict-last` to determine the end point of the element most recently added to a line; the main effect is that closing parentheses are attached in the right place when a multi-line `pict` is embedded in `stx`.

An identifier that starts with `_` is italicized in the `pict`, and the `_` is dropped, unless the `code-italic-underscore-enabled` parameter is set to `#f`. Also, unless `code-scripts-enabled` is set to `#f`, `_` and `~` in the middle of a word create superscripts and subscripts, respectively (like TeX); for example `foo~4_ok` is displayed as the identifier `foo` with a 4 superscript and an `ok` subscript.

Further, uses of certain identifiers in `stx` typeset specially:

- `code:blank` — produces a space.

- `(code:comment s ...)` — produces a comment block, with each *s* on its own line, where each *s* must be a string or a pict.
- `(code:line datum ...)` — typesets the *datum* sequence, which is mostly useful for the top-level sequence, since `typeset-code` accepts only one argument.
- `(code:contract datum ...)` — like `code:line`, but every *datum* is colored as a comment, and a `;` is prefixed to every line.
- `(code:template datum ...)` — like `code:line`, but a `;` is prefixed to every line.
- `$` — typesets as a vertical bar (for no particularly good reason).

`(code datum ...)`

The macro form of `typeset-code`. Within a *datum*, `unsyntax` can be used to escape to an expression, and identifiers bound as `syntax` to code transformers trigger transformations.

For more information, see `typeset-code` and `define-code`, since `code` is defined as

```
(define-code code typeset-code)
```

Examples:

```
> (code (+ 1 2))
(+ 1 2)
> (code (+ 1 #,(+ 1 1)))
(+ 1 2)
> (code (+ 1 #,(frame (code 2))))
(+ 1 2)
> (define-syntax two (make-code-transformer #'(code 2)))

> (code (+ 1 two))
(+ 1 2)
```

```
(current-code-font) → text-style/c
(current-code-font style) → void?
style : text-style/c
```

Parameter for a base font used to typeset text. The default is `(bold . modern)`. For even more control, see `current-code-tt`.

```
(current-code-tt) → (string? . -> . pict?)
(current-code-tt proc) → void?
proc : (string? . -> . pict?)
```

Parameter for a one-argument procedure to turn a string into a pict, used to typeset text. The default is

```
(lambda (s) (text s (current-code-font) ((get-current-code-font-size))))
```

This procedure is not used to typeset subscripts or other items that require font changes, where `current-code-font` is used directly.

```
(get-current-code-font-size) → (-> exact-nonnegative-integer?)  
(get-current-code-font-size proc) → void?  
proc : (-> exact-nonnegative-integer?)
```

A parameter used to access the default font size. The `slideshow/code` library initializes this parameter to `current-font-size`.

```
(current-code-line-sep) → real?  
(current-code-line-sep amt) → void?  
amt : real?
```

A parameter that determines the spacing between lines of typeset code. The default is 2.

```
(current-comment-color) → (or/c string? (is-a?/c color%))  
(current-comment-color color) → void?  
color : (or/c string? (is-a?/c color%))
```

A parameter for the color of comments.

```
(current-keyword-color) → (or/c string? (is-a?/c color%))  
(current-keyword-color color) → void?  
color : (or/c string? (is-a?/c color%))
```

A parameter for the color of syntactic-form names. See `current-keyword-list`.

```
(current-id-color) → (or/c string? (is-a?/c color%))  
(current-id-color color) → void?  
color : (or/c string? (is-a?/c color%))
```

A parameter for the color of identifiers that are not syntactic form names or constants.

```
(current-literal-color) → (or/c string? (is-a?/c color%))  
(current-literal-color color) → void?  
color : (or/c string? (is-a?/c color%))
```

A parameter for the color of literal values, such as strings and numbers. See also `current-literal-list`

```
(current-const-color) → (or/c string? (is-a?/c color%))
(current-const-color color) → void?
  color : (or/c string? (is-a?/c color%))
```

A parameter for the color of constant names. See `current-const-list`.

```
(current-base-color) → (or/c string? (is-a?/c color%))
(current-base-color color) → void?
  color : (or/c string? (is-a?/c color%))
```

A parameter for the color of everything else.

```
(current-reader-forms) → (listof symbol?)
(current-reader-forms syms) → void?
  syms : (listof symbol?)
```

Parameter for a list of symbols indicating which built-in reader forms should be used. The default is `'quasiquote`. Remove a symbol to suppress the corresponding reader output.

```
(code-align pict) → pict?
  pict : pict?
```

Adjusts the ascent of `pict` so that its bottom aligns with the baseline for text; use this function when `pict` has no ascent.

```
(current-keyword-list) → (listof string?)
(current-keyword-list names) → void?
  names : (listof string?)
```

A list of strings to color as syntactic-form names. The default includes all of the forms provided by `racket/base` and all of the forms provided by `mzscheme`.

```
(current-const-list) → (listof string?)
(current-const-list names) → void?
  names : (listof string?)
```

A list of strings to color as constant names. The default is `null`.

```
(current-literal-list) → (listof string?)
(current-literal-list names) → void?
  names : (listof string?)
```

A list of strings to color as literals, in addition to literals such as strings. The default is `null`.



```
| racket/base-const-list : (listof string?)
```

A list of strings that could be used to initialize the `current-const-list` parameter.

```
| mzscheme-const-list : (listof string?)
```

A list of strings that could be used to initialize the `current-const-list` parameter.

```
| (code-colorize-enabled) → boolean?  
| (code-colorize-enabled on?) → void?  
| on? : any/c
```

A parameter to enable or disable all code coloring. The default is `#t`.

```
| (code-colorize-quote-enabled) → boolean?  
| (code-colorize-quote-enabled on?) → void?  
| on? : any/c
```

A parameter to control whether under a quote is colored as a literal (as in this documentation). The default is `#t`.

```
| (code-italic-underscore-enabled) → boolean?  
| (code-italic-underscore-enabled on?) → void?  
| on? : any/c
```

A parameter to control whether `_`-prefixed identifiers are italicized (dropping the `_`). The default is `#t`.

```
| (code-scripts-enabled) → boolean?  
| (code-scripts-enabled on?) → void?  
| on? : any/c
```

A parameter to control whether TeX-style subscripts and superscripts are recognized in an identifier.

```
| (define-code code-id typeset-code-id)  
| (define-code code-id typeset-code-id escape-id)
```

Defines `code-id` as a macro that uses `typeset-code-id`, which is a function with the same input as `typeset-code`. The `escape-id` form defaults to `unsyntax`.

The resulting `code-id` syntactic form takes a sequence of *datums*:

```
(code-id datum ...)
```

It produces a pict that typesets the sequence. Source-location information for the *datum* determines the layout of code in the resulting pict. The *code-id* is expanded in such a way that source location is preserved during compilation (so *typeset-code-id* receives a syntax object with source locations intact).

If a *datum* contains (*escape-id expr*)—perhaps as #, *expr* when *escape-id* is un-syntax—then the *expr* is evaluated and the result datum is spliced in place of the *escape-id* form in *datum*. If the result is not a syntax object, it is given the source location of the *escape-id* form. A pict value intected this way as a *datum* is rendered as itself.

If a *datum* contains (*transform-id datum ...*) or *transform-id* for a *transform-id* that is bound as syntax to a code transformer, then the (*transform-id datum ...*) or *transform-id* may be replaced with an escaped expression, depending on the code transformer’s result.

```
(make-code-transformer proc-or-stx) → code-transformer?
  proc-or-stx : (or/c (syntax? . -> . (or/c syntax? #f))
                  syntax?)
prop:code-transformer : struct-type-property?
(code-transformer? v) → boolean?
  v : any/c
```

Exported for-syntax for creating *code transformers*.

For code transformer created with (*make-code-transformer proc*), *proc* takes a syntax object representing the use of an identifier bound to the transformer, and it produces an expression whose value replaces the identifier use within a code form or a form defined via *define-code*. Like a macro transformer, a code transformer is triggered either by a use of the bound identifier in an “application” position, in which case the transformer receives the entire “application” form, or the identifier by itself can also trigger the transformer. The code transformer’s *proc* can return #f, in which case the use of the identifier is left untransformed; if the identifier was used in an “application” position, the transformer *proc* will be called again for the identifier use by itself.

A code transformer produced by (*make-code-transformer stx*) is equivalent to

```
(make-code-transformer (lambda (use-stx)
  (if (identifier? use-stx)
      stx
      #f)))
```

A structure type with the *prop:code-transformer* property implements a code transformer. The property value must be a procedure of one argument, which receives the structure and returns a procedure that is like a *proc* passed to *make-code-transformer*, except that the property value takes the structure instance as an argument before the syntax object to transform.

The `code-transformer?` predicate returns `#t` for a value produced by `make-code-transformer` or for an instance of a structure type with the `prop:code-transformer` property, `#f` otherwise.

Example:

```
> (let-syntax ([bag (make-code-transformer #'(code hat))]
              [copy (make-code-transformer (syntax-rules ()
                                           [(_ c) (code (* 2 c))])])])
  (inset (frame (code ((copy cat) in the bag)) 2))
  (((* 2 cat) in the hat))
```

```
(define-exec-code (pict-id runnable-id string-id)
  datum ...)
```

Binds `pict-id` to the result of `(code datum ...)`, except that if an identifier `_` appears anywhere in a `datum`, then the identifier and the following expression are not included for code.

Meanwhile, `runnable-id` is bound to a syntax object that wraps the `datums` in a `begin`. In this case, `_s` are removed from the `datums`, but not the following expression. Thus, an `_` identifier is used to comment out an expression from the pict, but have it present in the syntax object for evaluation.

The `string-id` is bound to a string representation of the code that is in the pict. This string is useful for copying to the clipboard with `(send the-clipboard set-clipboard-string string-id 0)`.

```
(define-exec-code/scale scale-expr (pict-id runnable-id string-id)
  datum ...)
```

Like `define-exec-code`, but with a `scale` to use via `scale/improve-new-text` when generating the pict.

```
comment-color : (or/c string? (is-a?/c color%))
keyword-color : (or/c string? (is-a?/c color%))
id-color      : (or/c string? (is-a?/c color%))
literal-color : (or/c string? (is-a?/c color%))
```

For backward compatibility, the default values for `current-comment-color`, etc.

```
(code-pict-bottom-line-pict pict) → (or/c pict? #f)
pict : pict?
```

The same as `pict-last`, provided for backward compatibility.

```
(pict->code-pict pict bl-pict) → pict?  
  pict : pict?  
  bl-pict : (or/c pict? #f)
```

Mainly for backward compatibility: returns (if *bl-pict* (use-last pict (or (pict-last *bl-pict*) *bl-pict*))).

## 8 Animation Helpers

These functions are designed to work with the slide constructors in `slideshow/play`.

### 8.1 Pict Interpolations

```
(fade-pict n p1 p2 [#:combine combine]) → pict?  
  n : (real-in 0.0 1.0)  
  p1 : pict?  
  p2 : pict?  
  combine : (pict? pict? . -> . pict?) = cc-superimpose
```

Interpolates *p1* and *p2*, where the result with *n* as 0.0 is *p1*, and the result with *n* as 1.0 is *p2*. For intermediate points, *p1* fades out while *p2* fades in as *n* changes from 0.0 to 1.0. At the same time, the width and height of the generated pict are intermediate between *p1* and *p2*, and the relative baselines and last pict correspondingly morph within the bounding box.

The *combine* argument determines how *p1* and *p2* are aligned for morphing. For example, if *p1* and *p2* both contain multiple lines of text with the same line height but different number of lines, then using `ctl-superimpose` would keep the ascent line in a fixed location relative to the top of the resulting pict as the rest of the shape morphs around it.

```
(fade-around-pict n p1 make-p2) → pict?  
  n : (real-in 0.0 1.0)  
  p1 : pict?  
  make-p2 : (pict? . -> . pict?)
```

Similar to `fade-pict`, but the target is not a fixed *p2*, but instead a function *make-p2* that takes a `laundered ghost` of *p1* and places it into a larger scene. Also, *p1* does not fade out as *n* increases; instead, *p1* is placed wherever its ghost appears in the result of *make-p2*.

For example,

```
(lambda (n)  
  (fade-around-pict n  
                    (code x)  
                    (lambda (g) (code (+ #,x 1))))))
```

animates the wrapping of *x* with a `(+ ... 1)` form.

```
(slide-pict base p p-from p-to n) → pict?  
  base : pict?
```

```

p : pict?
p-from : pict?
p-to : pict?
n : (real-in 0.0 1.0)

```

Pins  $p$  onto  $base$ , sliding from  $p-from$  to  $p-to$  (which are `picts` within  $base$ ) as  $n$  goes from `0.0` to `1.0`. The top-left locations of  $p-from$  and  $p-to$  determine the placement of the top-left of  $p$ .

The  $p-from$  and  $p-to$  `picts` are typically `laundered ghosts` of  $p$  within  $base$ , but they can be any `picts` within  $base$ .

```

(slide-pict/center base p p-from p-to n) → pict?
base : pict?
p : pict?
p-from : pict?
p-to : pict?
n : (real-in 0.0 1.0)

```

Like `slide-pict`, but aligns the center of  $p$  with  $p-from$  and  $p-to$ .

## 8.2 Merging Animations

```

(sequence-animations gen ...) → (-> (real-in 0.0 1.0) pict?)
gen : (-> (real-in 0.0 1.0) pict?)

```

Converts a list of `gen` functions into a single function that uses each `gen` in sequence.

```

(reverse-animations gen ...) → (-> (real-in 0.0 1.0) pict?)
gen : (-> (real-in 0.0 1.0) pict?)

```

Converts a list of `gen` functions into a single function that run `(sequence-animations gen ...)` in reverse.

## 8.3 Stretching and Squashing Time

```

(fast-start n) → (real-in 0.0 1.0)
n : (real-in 0.0 1.0)
(fast-end n) → (real-in 0.0 1.0)
n : (real-in 0.0 1.0)
(fast-edges n) → (real-in 0.0 1.0)
n : (real-in 0.0 1.0)

```

```
(fast-middle n) → (real-in 0.0 1.0)
n : (real-in 0.0 1.0)
```

Monotonically but non-uniformly maps  $n$  with fixed points at 0.0 and 1.0.

The `fast-start` mapping is convex, so that

```
(slide-pict base p p1 p2 (fast-start n))
```

appears to move quickly away from  $p1$  and then slowly as it approaches  $p2$ , assuming that  $n$  increases uniformly.

The `fast-end` mapping is concave, so that

```
(slide-pict base p p1 p2 (fast-end n))
```

appears to move slowly away from  $p1$  and then quickly as it approaches  $p2$ , assuming that  $n$  increases uniformly.

The `fast-edges` mapping is convex at first and concave at the end, so that

```
(slide-pict base p p1 p2 (fast-edges n))
```

appears to move quickly away from  $p1$ , then more slowly, and then quickly again near  $p2$ , assuming that  $n$  increases uniformly.

The `fast-middle` mapping is concave at first and convex at the end, so that

```
(slide-pict base p p1 p2 (fast-middle n))
```

appears to move slowly away from  $p1$ , then more quickly, and then slowly again near  $p2$ , assuming that  $n$  increases uniformly.

```
(split-phase n) → (real-in 0.0 1.0) (real-in 0.0 1.0)
n : (real-in 0.0 1.0)
```

Splits the progression of  $n$  from 0.0 to 1.0 into a progression from (`values 0.0 0.0`) to (`values 1.0 0.0`) and then (`values 1.0 0.0`) to (`values 1.0 1.0`).

## 9 Tree Layout


These functions specify tree layouts and functions that render them as `picts`.

```
(require pict/tree-layout)      package: pict-lib

(tree-layout [#:pict node-pict] child ...) → tree-layout?
  node-pict : (or/c #f pict?) = #f
  child : (or/c tree-layout? tree-edge? #f)
```

Specifies an interior node of a tree for use with one of the renderers below.

If the children are `tree-layout?`s, then they have edges created by passing the corresponding `tree-layout?`s directly to `tree-edge`. Children that are `#f` correspond to leaf nodes that are not drawn.

The default `node-pict` (used when it is `#f`) is 


```
(tree-edge node
  [#:edge-color edge-color
   #:edge-width edge-width]) → tree-edge?
  node : tree-layout?
  edge-color : (or/c string?                = "gray"
               (is-a?/c color%))
               (list/c byte? byte? byte?))
  edge-width : (or/c 'unspecified real? #f) = 'unspecified
```

This function specifies an edge from some parent to the given `node`. It is intended to be used with `tree-layout`.

When `edge-width` is `'unspecified`, the line width will not be set. This is intended to allow the line width to be set for the whole pict via `linewidth`. Otherwise, `edge-width` is interpreted the same way as the width argument for the `linewidth` function.

Example:

```
> (naive-layered (tree-layout
  (tree-edge #:edge-width 3 (tree-layout))
  (tree-edge #:edge-color "green" (tree-layout))))
```



Changed in version 6.1.0.5 of package `pict-lib`: Added an `#:edge-width` option

```
(tree-layout? v) → boolean?
  v : any/c
```



Recognizes a tree layout. It returns `#t` when given `#f` or the result of `tree-layout`.

```
(binary-tree-layout? v) → boolean?  
v : any/c
```

Recognizes a `tree-layout?` that represents a binary tree. That is, each interior node has either two children or is `#f`. Note that a node with zero children does not count as a leaf for the purposes of `binary-tree-layout?`.

Examples:

```
> (binary-tree-layout? (tree-layout #f #f))  
#t  
> (binary-tree-layout? #f)  
#t  
> (binary-tree-layout? (tree-layout (tree-layout) (tree-layout)))  
#f
```

```
(tree-edge? v) → boolean?  
v : any/c
```


Recognizes an `tree-edge`.

```
(naive-layered tree-layout  
  [#:x-spacing x-spacing  
   #:y-spacing y-spacing]) → pict?  
tree-layout : tree-layout?  
x-spacing : (or/c (and/c real? positive?) #f) = #f  
y-spacing : (or/c (and/c real? positive?) #f) = #f
```


Uses a naive algorithm that ensures that all nodes at a fixed depth are the same vertical distance from the root (dubbed “layered”). It recursively lays out subtrees and then horizontally combines them, aligning them at their tops. Then it places the root node centered over the children nodes.

Examples:

```
> (define (complete d)  
  (cond  
    [(zero? d) #f]  
    [else (define s (complete (- d 1)))  
           (tree-layout s s)]))  
  
> (naive-layered (complete 4))
```




```
> (naive-layered (tree-layout
                 (tree-layout)
                 (tree-layout)
                 (tree-layout)
                 (tree-layout)
                 (tree-layout)
                 (tree-layout)
                 (tree-layout)
                 (tree-layout))))
```



```
> (define right-subtree-with-left-chain
    (tree-layout
      (tree-layout
        (tree-layout #f #f)
        (tree-layout
          (tree-layout #f #f)
          #f))
      (tree-layout
        (tree-layout
          (tree-layout
            (tree-layout #f #f)
            #f)
          #f)
        #f)
      #f)))
```

```
> (naive-layered right-subtree-with-left-chain)
```



```
(binary-tidier tree-layout
  [#:x-spacing x-spacing
   #:y-spacing y-spacing]) → pict?
tree-layout : binary-tree-layout?
x-spacing : (or/c (and/c real? positive?) #f) = #f
y-spacing : (or/c (and/c real? positive?) #f) = #f
```

Uses the layout algorithm from *Tidier Drawing of Trees* by Edward M. Reingold and John S. Tilford (IEEE Transactions on Software Engineering, Volume 7, Number 2, March 1981) to lay out *tree-layout*.

The layout algorithm guarantees a number of properties, namely:

- nodes at the same level of tree appear at the same vertical distance from the top of the pict
- parents are centered over their children, which are placed from left to right,
- isomorphic subtrees are drawn the same way, no matter where they appear in the complete tree, and
- a tree and its mirror image produce pict's that are mirror images of each other (which also holds for subtrees of the complete tree).

Within those constraints, the algorithm tries to make as narrow a drawing as it can, even to the point that one subtree of a given node might cross under the other one.


More precisely, it recursively lays out the two subtree and then, without adjusting the layout of the two subtrees, moves them as close together as it can, putting the root of the new tree centered on top of its children. (It does this in linear time, using clever techniques as discussed in the paper.)

The *x-spacing* and *y-spacing* are the amount of space that each row and each column takes up, measured in pixels. If *x-spacing* is #f, it is the width of the widest node *pict?*

in the tree. If *y-spacing* is #f, it is 1.5 times the width of the widest node `pict?` in the tree.

Examples:


```
> (binary-tidier (complete 4))
```




```
> (define (dl t) (tree-layout (tree-layout #f #f) t))
```

```
> (define (dr t) (tree-layout t (tree-layout #f #f)))
```

```
> (binary-tidier
  (tree-layout
    (dr (dr (dr (dl (dl (dl (complete 2)))))))
    (dl (dl (dl (dr (dr (dr (complete 2))))))))))
```



```
> (binary-tidier right-subtree-with-left-chain)
```



```

(hv-alternating tree-layout
  [#:x-spacing x-spacing
   #:y-spacing y-spacing]) → pict?
tree-layout : binary-tree-layout?
x-spacing : (or/c (and/c real? positive?) #f) = #f
y-spacing : (or/c (and/c real? positive?) #f) = #f


```

Uses the “CT” binary tree layout algorithm from *A note on optimal area algorithms for upward drawing of binary trees* by P. Crescenzi, G. Di Battista, and A. Piperno (Computational Geometry, Theory and Applications, 1992) to lay out *tree-layout*.

It adds horizontal and vertical space between layers based on *x-spacing* and *y-spacing*. If either is #f, 1.5 times the size of the biggest node is used.

Example:

```
> (hv-alternating (complete 8))
```



Added in version 6.0.1.4 of package `pi ct-lib`.

## 10 Miscellaneous

```
(hyperlinkize pict) → pict?  
pict : pict?
```

Adds an underline and blue color. The *pict*'s height and descent are extended.

```
(scale-color factor color) → (is-a?/c color%)  
factor : real?  
color : (or/c string (is-a?/c color%))
```

Scales a color, making it brighter or darker. If the factor is less than 1, the color is darkened by multiplying the RGB components by the factor. If the factor is greater than 1, the color is lightened by dividing the gap between the RGB components and 255 by the factor.

```
(color-series dc  
    max-step  
    step-delta  
    start  
    end  
    proc  
    set-pen?  
    set-brush?) → void?  
dc : (is-a?/c dc<%>)  
max-step : exact-nonnegative-integer?  
step-delta : (and/c exact? positive?)  
start : (or/c string? (is-a?/c color%))  
end : (or/c string? (is-a?/c color%))  
proc : (exact? . -> . any)  
set-pen? : any/c  
set-brush? : any/c
```

Calls a *proc* multiple times, gradually changing the pen and/or brush color for each call. For the first call, the current pen and/or brush color matches *start*; for the last call, it matches *end*; and for intermediate calls, the color is an intermediate color.

The *max-step* and *step-delta* arguments should be exact numbers; the procedure is called with each number from 0 to *max-step* inclusive using a *step-delta* increment.

## 11 Rendering

```
(dc-for-text-size) → (or/c #f (is-a?/c dc<%>))
(dc-for-text-size dc) → void?
  dc : (or/c #f (is-a?/c dc<%>))
```

A parameter that is used to determine the bounding box of pict's created with `text`.

The drawing context installed in this parameter need not be the same as the ultimate drawing context, but it should measure text in the same way. Under normal circumstances, font metrics are the same for all drawing contexts, so the default value of `dc-for-text-size` is a `bitmap-dc%` that draws to a 1-by-1 bitmap.

```
(convert-bounds-padding)
→ (list/c (>=/c 0) (>=/c 0) (>=/c 0) (>=/c 0))
(convert-bounds-padding padding) → void?
  padding : (list/c (>=/c 0) (>=/c 0) (>=/c 0) (>=/c 0))
```

A parameter that determines an amount of padding added to each edge of a pict when converting to a format like `'png@2x-bytes+bounds8` (see `file/convertible`). The default value of the parameter is `'(3 3 3 3)`, which adds three pixels to each edge to accommodate a small amount of drawing outside the pict's bounding box.

Added in version 1.2 of package `pict-lib`.

```
(draw-pict pict dc x y) → void?
  pict : pict?
  dc : (is-a?/c dc<%>)
  x : real?
  y : real?
```

Draws `pict` to `dc`, with its top-left corner at offset `(x, y)`.

```
(pict->bitmap pict [smoothing]) → (is-a?/c bitmap%)
  pict : pict?
  smoothing : (or/c 'unsmoothed 'smoothed 'aligned) = 'aligned
```

Returns a `bitmap%` with an alpha channel, no larger than `pict`, with `pict` drawn on it in the top-left corner `(0, 0)`.

When drawing the pict into the bitmap using the smoothing mode `smoothing` (see `dc<%>` for more information on smoothing modes).

```
(pict->argb-pixels pict [smoothing]) → bytes?
  pict : pict?
  smoothing : (or/c 'unsmoothed 'smoothed 'aligned) = 'aligned
```



Returns the `bytes?` with the pixels corresponding the bitmap that `pict->bitmap` returns. Each pixel has four bytes in the result: the alpha, red, green, and blue components.

Examples:

```
> (pict->argb-pixels
   (filled-rectangle 1 1))
#\377\0\0\0"
> (pict->argb-pixels
   (colorize (filled-rectangle 1 1) "red"))
#\377\377\0\0"
```

Added in version 1.1 of package `pict-lib`.

```
(argb-pixels->pict bytes width) → pict?
  bytes : bytes?
  width : exact-nonnegative-integer?
```

Constructs a `pict` from `bytes` with the width `width`. Each pixel in the resulting `pict` corresponds to four entries in `bytes`: the alpha value, and the red, green, and blue values.

Example:

```
> (let ([b (make-bytes (* 40 40 4) 255)])
     (for ([x (in-range (bytes-length b))])
         ; when in one of two vertical bands (10-20 & 30-40)
         (when (or (<= 10 (modulo (quotient x 4) 40) 20)
                   (<= 30 (modulo (quotient x 4) 40) 40))
                 ; change the red and green fields of the pixel
                 (when (= 1 (modulo x 4)) (bytes-set! b x 0))
                 (when (= 2 (modulo x 4)) (bytes-set! b x 150))))
       (argb-pixels->pict b 40))
```



Added in version 1.1 of package `pict-lib`.

```
(make-pict-drawer pict)
→ ((is-a?/c dc<%>) real? real? . -> . void?)
  pict : pict?
```

Generates a `pict-drawer` procedure for multiple renderings of `pict`. Using the generated procedure can be faster than repeated calls to `draw-pict`.

```

(show-pict pict
  [w
   h]
  #:frame-x frame-x
  #:frame-y frame-y
  #:frame-style frame-style) → void?
pict : pict?
w : (or/c #f exact-nonnegative-integer?) = #f
h : (or/c #f exact-nonnegative-integer?) = #f
frame-x : (or/c (integer-in -10000 10000) #f)
frame-y : (or/c (integer-in -10000 10000) #f)
frame-style : (listof (or/c 'no-resize-border 'no-caption
                            'no-system-menu 'hide-menu-bar
                            'toolbar-button 'float 'metal))

```

Opens a frame that displays *pict*. The frame adds one method, *set-pict*, which takes a pict to display. The optional *w* and *h* arguments specify a minimum size for the frame's drawing area, and the *frame-x*, *frame-y*, and *frame-style* keyword arguments behave in the same manner as *x*, *y*, and *style* arguments for the *frame%*.

```

(current-expected-text-scale) → (list/c real? real?)
(current-expected-text-scale scales) → void?
scales : (list/c real? real?)

```

A parameter used to refine text measurements to better match an expected scaling of the image. The *scale/improve-new-text* form sets this parameter while also scaling the resulting pict.

## 12 Conversion to Picts

```
(require pict/convert)    package: pict-lib
```

The `pict/convert` library defines a protocol for values to convert themselves to pict. The protocol is used by DrRacket's interactions window, for example, to render values that it prints.

```
| prop:pict-convertible : struct-type-property?
```

A property whose value should be a procedure matching the contract `(-> any/c pict?)`. The procedure is called when a structure with the property is passed to `pict-convert`; the argument to the procedure is the structure, and the procedure's result should be a pict.

```
| prop:pict-convertible? : struct-type-property?
```

A property whose value should be a predicate procedure (i.e., matching the contract `predicate/c`).

If this property is not set, then it is assumed to be the function `(λ (x) #t)`.

If this property is set, then this procedure is called by `pict-convertible?` to determine if this particular value is convertible (thereby supporting situations where some instances of a given struct are convertible to pict, but others are not).

```
| (pict-convertible? v) → boolean?  
  v : any/c
```

Returns `#t` if `v` supports the conversion protocol (by being a struct with the `prop:pict-convertible` property) and `#f` otherwise.

```
| (pict-convert v) → pict?  
  v : pict-convertible?
```

Requests a data conversion from `v` to a pict.